Tutorial 1: Simple Instructions to Move Data

New Instructions Introduced

LD
LDM

ST
STM

STL
STH

Overview of Tutorial

The first tutorial of this series looks at a fundamental operation, that of moving data between data memory (RAM) and the two working registers named accumulator A and accumulator B. 

Loading the Accumulators

Let us start with a simple operation, that of assignment. In the C language we can assign a value to a 32-bit variable as follows:

Example 1-1.

	
	
	

	unsigned long A = 1;
	// Load a 32-bit variable A with 1
	


The TMS320C5000 has two 40-bit working registers called accumulator A and accumulator B. For the purpose of this tutorial, we shall treat them as 32-bit registers. 

In order to load either accumulator A or accumulator B with a constant, as we have already seen in Example 1-1, we use the instruction LD (load accumulator): 

Example 1-2.

	
	
	

	LD #1, A


	; Load the constant 1 into accumulator A.  

; Accumulator A now contains 00000001

; hexadecimal.
	


The instruction LD (load accumulator) needs two additional pieces of information known as operands. 

The first operand uses the symbol # to indicate that the value is a constant. 

The second operand is referred to as the destination. By the term destination we mean where we are going to load the data. With the instruction  LD (load accumulator) we have the choice of either accumulator A or accumulator B as the destination. In Example 1-2 we have used accumulator A as the destination.

Should we wish to load the constant 255 into accumulator B we would write: 

Example 1-3.

	
	
	

	LD #255, B


	; Load the constant 255 decimal into

; accumulator B. 
	


Technically, using the instruction LD (load accumulator) with a constant is known as immediate addressing. The value following immediately after the symbol # is the constant to be loaded.   

The second operand supplying the name of the destination must be either accumulator A or accumulator B. To omit the accumulator type or to use a negative constant are both incorrect, as shown in Example 1-4. 

Example 1-4.

	
	
	

	LD #33
	; Incorrect. Either accumulator A or

; accumulator B must be specified as the

; destination.
	

	LD #-1, A
	; Incorrect. The number must be positive.
	


To represent a constant expressed as a hexadecimal number we write the letter h after the number. The letter h may be in upper case in lower case. 

Example 1-5.

	
	
	

	LD #64h, A


	; Load the constant 64h (100 decimal) into

; accumulator A. Accumulator A now contains

; 00000064h. 
	


It is illegal to start the first number of the constant with one of the hexadecimal characters  A,B,C,D,E or F. These must be preceded by a 0, otherwise the assembler will take them to be labels rather than hexadecimal numbers. 

Example 1-6.

	
	
	

	LD #3Fh, B
	; Correct. Letter F preceded by the number 3.
	

	LD #ACh, B
	; Incorrect. Letter A must be preceded by 0.
	

	LD #0ACh,A
	; Correct. Letter A is preceded by the number

; 0.
	


If the letter h is accidentally left out, then the assembler treats the numbers as being of a different format:

Example 1-7.

	
	
	

	LD #33, A


	; Load accumulator A with the constant 33

; decimal, rather than the constant 33

; hexadecimal as was intended.
	

	LD #0AA,B
	; Error. “Illegal octal constant”.
	


Because the constant can be up to 16 bits long, the range of numbers we may load using the instruction LD (load accumulator) is 0 to 65535 (0 to FFFFh). Some samples of loading the accumulators are given in Example 1-8: 

Example 1-8.

	
	
	

	LD #1000, B


	; Load accumulator B with the 16-bit 

; constant 1000 decimal. Accumulator B now ; contains 000003E8h.
	

	LD #0h, B


	; Clear accumulator B. Accumulator B now

; contains 00000000h.
	

	LD #0FFFFh, A


	; Load accumulator A with maximum allowed

; 16-bit constant FFFFh (65535 decimal).

; Accumulator A now contains 0000FFFFh.
	


In Example 1-8 we have used the term clear to mean load the variable with 0. 

Loading Auxiliary Registers with Constants

As part of its architecture, the TMS320C5000 has eight auxiliary registers named AR0, AR1, AR2, AR3, AR4, AR5, AR6 and AR7. These are used for general purpose, in particular for counters and accessing data memory. 

The auxiliary registers AR0 to AR7 form part of what are known as memory-mapped registers. Within this group are the timers, ports and auxiliary registers AR0 to AR7. Here memory-mapped means that the registers are treated as a special series of 16-bit data memory addresses. For the time being, the only memory-mapped registers we will concern ourselves with are the auxiliary registers AR0 to AR7.  

We can load a 16-bit constant into one of the memory-mapped registers. For this we use the instruction STM (store immediate value into memory-mapped register).

Example 1-9.

	
	
	

	STM #199, AR3


	; Store the 16-bit constant 199 decimal

; in the memory-mapped register AR3.

; AR3 now contains 199 decimal, which

; in hexadecmial is is 00C7h. 
	

	STM #2245h, AR7


	; Store the 16-bit constant 2245h in

; the memory-mapped register AR7.
	


The operand following immediately after the symbol # is the 16-bit constant. Here the second operand is the destination (i.e. where we are going to store the constant), and is  one of the auxiliary registers AR0 to AR7. Two operands are required, so the following is illegal: 

Example 1-10.

	
	
	

	STM #44h


	; Illegal. Missing operand. An auxiliary

; register AR0 to AR7 must be specified.
	


When the assembler encounters the operands AR0 to AR7, it converts them to the  numbers 10h to 17h, which represent the addresses in the memory-map. The following is therefore legal, but not recommended: 

Example 1-11.

	
	
	

	STM #44h, 10h


	; Load constant 44h into auxiliary register

; AR0. Legal but not recommended.
	


There are two problems with Example 1-11. First, it is not immediately obvious that 10h corresponds to auxiliary register AR0. Anything not obvious increases the likelihood of making mistakes. Second, there is the issue of compatibility. Should a later version of the TMS320C5000 use a slightly different memory-mapping, the wrong memory-mapped addresses will be used. Hence, the code will not be completely compatible. 

Direct Addressing

Normally in a program, the accumulators A and B would be used as working registers. The auxiliary registers AR0 to AR7 would usually be used for general purpose, for loop counters and to access data memory. In order to store more variables, we need access to more data memory addresses. For this we use a slightly different version of the instruction LD (load accumulator).

To load accumulator A or accumulator B with the data from a data memory address, we can write:

Example 1-12.

	
	
	

	LD 60h, A


	; Load accumulator A with the value

; contained at data memory address 60h.  
	

	LD 61h, B


	; Load accumulator B with the value

; contained at data memory address 61h.
	


This is known as direct addressing. We load either accumulator A or accumulator B directly with the contents of a specified data memory address. When we load an accumulator, we in fact make a copy of the contents of the data memory address and put the copy into the accumulator. The original contents of the data memory address remain unchanged.   

Note that the syntax of immediate addressing (with the symbol #) is different to that of direct addressing (without the symbol #). A very easy mistake to make when using indirect addressing is to forget to type the symbol # and this completely changes the meaning of the instruction:

Example 1-13.

	
	
	

	LD #10h, A


	; Load the constant 10h into accumulator A

; using immediate addressing.
	

	LD 10h, A


	; Erroneous attempt to load the constant

; 10h into accumulator A. In fact loads

; the contents of data memory address

; 10h into accumulator A using direct 

; addressing.
	


The operand providing the direct address for the instruction LD must lie in the range 0 to 127 (7Fh). Any value outside this range will be taken as an error, as shown in Example 1-14:

Example 1-14.

	
	
	

	LD -1, A
	; Illegal. Operand less than minimum allowed. 
	

	LD 320h, B


	; Illegal. Operand greater than maximum

; allowed.
	


In the Texas Instrumetns databook, the term data memory address is often abbreviated to dmad.

Copying the Contents of an Auxiliary Register to an Accumulator

Since auxiliary registers AR0 to AR7 are treated as memory-mapped registers, we can load a value from one of auxiliary registers AR0 to AR7 into either accumulator A or accumulator B using the instruction LDM (load from memory-mapped register):

Example 1-15.

	
	
	

	LDM  AR0, A


	; Load the contents of auxiliary register

; AR0 into accumulator A.
	

	LDM  AR7, B
	; Load the contents of auxiliary register 

; AR7 into accumulator B. 
	


The instruction LDM (load from memory-mapped register) takes two operands. The first operand is one of the auxiliary registers AR0 to AR7. The second operand is the destination and can be accumulator A or accumulator B.

The Structure of Data Memory

Care must be taken when using direct addressing. Each data memory address is 16 bits, and hence lies in the range 0 to 65535 (0 to FFFFh). However, the operand used with the instruction LD provides only the lowest 7 bits of this address, and hence lies in the range 0 to 127  (0 to 7Fh). In short, the operand used for direct addressing only specifies part of the address. So what provides the other 16 - 7 = 9 bits of the data memory address?   

The remaining 9 bits are provided from another source known as the data memory page pointer (DP). 

At power up, the data memory page pointer (DP) is not initialized to any particular value, so one of the first instructions in any program must be to configure it. If not initialized before use, the instruction LD cannot be guaranteed to work in a repeatable manner. To initialize the data memory page pointer (DP) we use a variation of the instruction LD.


Example 1-16.

	
	
	

	LD #0, DP


	; Load data memory page pointer (DP) with a 

; 9-bit constant. The constant must lie in the

; range 0 to 511. For the TMS320C5000 use a

; value in the range 0 to 39 decimal for

; access to internal data memory addresses 0h

; to 13FFh.
	


If we look at the memory map of the TMS320C541 (the member of the TMS320C54x family with the least data memory available), we see that the data memory locations are arranged in pages as follows:

Table 1-1.
Data Memory Addresses of the TMS320C541

	Pages
	Available 

Addresses (Decimal)
	Available Addresses (Hexadecimal)
	

	0


	 96  to 127


	60h  to 7Fh


	

	1


	128 to 255


	80h to FFh


	

	2


	256 to 383


	100h to 17Fh


	

	3 to 38


	384 to 4991 


	180h to 137Fh


	

	39


	 4992 to 5119 


	1380h to 13FFh
	


If using a device other than the TMS320C541, it is recommended that the databook be checked in order to determine the range of valid data memory addresses. 

Note that in Page 0, data memory addresses 0 to 5Fh are reserved for memory-mapped registers and are therefore unavailable as general purpose data memory.  

Therefore, after we execute the following instruction

	LD #0, DP
	; Load data memory page pointer with zero.


 the only addresses available for general purpose data are 60h to 7Fh. 

As another example,  using the instruction LD #4, DP gives us access to page 4 addresses 200h to 27Fh. In order to access external date memory we would use pages in the range 40 to 511. 

Calculating Data Memory Addresses

When using direct addressing, in order to carry out an operation on a particular data memory address, we must first determine which memory page to use. 

Say we wish to use direct addressing to load the accumulator with the contents of data memory address 310h. To calculate the page number we need to divide 310h (784) by 80h (128) and round down to the nearest whole number. This gives us page number 6, which starts at address 718 (300h). We then add the offset supplied by the operand. 

Example 1-17.

	
	
	

	LD #6, DP


	; Set data memory page pointer to page 6. This

; gives us access to data memory addresses

; 300h to 37Fh.
	

	LD 10h, A


	; Load accumulator A with the contents of data

; memory address 300h + 10h = 310h.
	


A point of caution. Table 1-1 is for the TMS320C541 and if a different device is being used, the data memory addresses available for general purpose may be at different addresses on the memory map. It is therefore important to check the actual memory map for the device being used. 

Moving Data Words to Data Memory 

The instruction LD (load) is used to move a value to either accumulator A or accumulator B. To move a value to a data memory address we use variations on the instruction ST (store).

We can store an immediate value at a data memory address using the instruction ST (store). For example, to store the immediate value 100h at data memory address 3A0h we would write code as shown in Example 1-18:

Example 1-18.

	
	
	

	LD #7, DP


	; Page 7. Gain access to data memory   

; addresses 380h to 3FFh.
	

	ST #100h, 20h


	; Store the immediate value 100h at data   ; memory address 380h + 20h = 3A0h.  
	


To move data from an accumulator to a data memory address we use one of the store instructions STH (store accumulator high in data memory) and STL (store accumulator low in data memory).

Say we wish to copy the contents of accumulator A to data memory addresses 3A0h and 3A1h.  For this we need to select page 7.

Example 1-19.

	
	
	

	LD #7, DP


	; Page 7. Gain access to data memory   

; addresses 380h to 3FFh.
	

	STH A, 20h


	; Store high 16 bits of accumulator A in data  ; memory address 380h + 20h = 3A0h.  
	

	STL A, 21h


	; Store low 16 bits of accumulator A in data

; memory address 380h + 21h = 3A1h. 
	


The contents of the accumulator remain unchanged. We can also copy the contents of the accumulator to data memory by saving the low word then the high word: 

Example 1-20.

	
	
	

	LD #7, DP


	; Page 7. Gain access to data memory   

; addresses 380h to 3FFh.
	

	STL A, 20h


	; Store low 16 bits of accumulator A in data   ; memory address 380h + 20h = 3A0h.  
	

	STH A, 21h


	; Store high 16 bits of accumulator A in data

; memory address 380h + 21h = 3A1h. 
	


The only difference between the two orders of storage is one of convention. By loading the low word into the low address and the high word into the high address, we maintain compatibility with other instructions and the Texas Instruments TMS320C54x C compiler.

Upgrading from the TMS320C2000 to the TMS320C5000 

The TMS320C5000 has a different architecture to that of the earlier Texas Instruments DSP devices in that it has two accumulators (A and B), rather than just one. Therefore new instructions have been added that support two accumulators. This means the earlier instructions which only support one accumulator cannot be used. 

For each of the instructions used by the TMS320C000 there is an equivalent instruction for the TMS320C5000x, as given in Table 1-2:  

Table 1-2: Comparison of Instructions

	Description
	TMS320C2000 Instruction
	TMS320C5000 

Instruction
	

	Load 8-bit immediate
	LACL #x
	LD #x, Acc
	

	Load 16-bit immediate
	LACL #x
	LD #x, Acc
	

	Load direct
	LACC dmad
	LD dmad 
	

	Load auxiliary register 8-bit
	LAR ARx


	STM #x, ARx


	

	Load auxiliary register 16-bit
	LAR ARx


	STM #x, ARx


	

	Set data page
	LDP #x
	LD #x, DP
	

	Store accumulator low 
	SACL 
	STL Acc 
	

	Store accumulator high
	SACH 
	STH Acc
	

	Store immediate in data memory 
	LAMM #x, dmad

	ST #x, dmad

	


Here the term x is used to represent a constant, dmad represents a data memory address, ARx is used to denote one of the auxiliary registers AR0 to AR7 and Acc represents accumulator A or B.

Load and store instructions on the TMS320C5000 require two operands, whereas the earlier devices only required one. The second operand is used to specify either accumulator A or accumulator B.

On the TMS320C2000, auxiliary registers AR0 to AR7 are treated as special cases and hence have their own instructions LAR, LARK and LRLK. The TMS320C5000 has simplified matters by making the auxiliary registers part of the memory-mapped registers. 

Operations on the auxiliary registers AR0 to AR7 on the TMS320C5000 are done in a completely different way. In fact, a load instruction on the TMS320C2000 has been replaced with a store instruction on the TMS320C5000.

An example of an instruction to load an auxiliary register on the TMS320C2000 is shown in Example 1-21:

Example 1-21.

	
	
	

	LARK AR0, 10h


	; Load auxiliary register AR0 with

; constant 10h.
	


The same operation as carried out in Example 1-21 is shown using the TMS320C5000 instruction in Example 1-22:

Example 1-22.

	
	
	

	STM #10h, AR0


	; Store immediate value 10h in memory -

; mapped register AR0.
	


Besides the different instructions to carry out operations on auxiliary registers, a further point of unfamiliarity is that the order of operands is reversed. Care also needs to be taken when first using the TMS320C5000 to remember to type in the symbol # 
when using immediate data. If not, direct data will be loaded.

On the TMS320C2000, the assembler assigns the values 0 to 7 to the auxiliary registers AR0 to AR7. This means we can write:

Example 1-23.

	
	
	

	LARK 0, 10
	; Load AR0 with constant 10.
	


This should not be used on the TMS320C5000 because the memory mapping is different.  On the TMS320C5000, the auxiliary registers AR0 to AR7 are memory-mapped to locations 10h to 17h. In the following example, using the TMS320C5000 mnemonic with reversed operands does not work as expected.

Example 1-24.

	
	
	

	LARK 0, 10h
	; Syntax for earlier devices. Load

; auxiliary register AR0 with 10h.
	

	STM #10h, 0


	; Syntax for TMS320C5000. Unsuccessful

; attempt to load AR0 with the immediate

; value 10h. In fact loads the immediate 

; value 10h into memory-mapped register 0

; which is the interrupt mask register

; (IMR)!
	


There is one slight change in terminology introduced on the TMS320C5000. The term data memory address is abbreviated to dmad. It used to be abbreviated to dma.
Questions

	1.
	What is meant by the term destination?
	

	2.
	What is meant by the term immediate addressing?
	

	3.


	The letter h is used as an abbreviation for:

a)   hexadecimal

b) 
high

c) 
home
	

	4.
	Which two of the following immediate addressing instructions are legal?
	

	
	a) LD #FF, A

b) LD #70h, B

c) LD #0E0h, A

d) LD #66h, C

e) LD #222h

f) LD #0E3, B
	

	5.
	Write code to store the value 100h in auxiliary register AR6.
	

	6.
	What is the difference between the following?

a) 
LD #7Fh, B

b) 
LD 7Fh, B
	

	7.
	Which two of the following instructions are legal?

a) 
LD 123h, A

b) 
LD -23, B

c) 
LD 28h, A
d) 
LD 127, B
e) 
LD 128, C

f) 
LD 14h
	

	8.
	What does the abbreviation dmad stand for?
	

	9.


	Why do we need to initialize the data memory page pointer (DP) soon after reset?
	

	10. 


	Why do we only use data memory addresses 60h to 7Fh on page 0 for

general purpose?
	

	11.


	Using direct addressing, how would we write 1000h to data memory address 230h?
	

	12.
	Why would we use the instructions STL then STH in that order?
	

	13.
	What is the difference between the instructions LD (load) and ST(store)?

